

Obrázek 3.8: Část výsledné kuželové plochy

3.4 Řešení Příkladu 1 (str.55) v programu Maple

Zadání: Vyšetřete kvadriku [2], [5]

$$7x^2 + 6y^2 + 5z^2 - 4xy - 4yz - 22x + 24y + 2z + 30 = 0. \quad (3.1)$$

Kompletní kód řešení v Maple:

Poznámka: Výstupy některých příkazů kódu nejsou v textu uvedeny. Většinou se jedná o případy, kdy výstup pouze kopíruje zadaný příkaz.

```
> restart;
> with(LinearAlgebra): with(linalg): with(plots):
```

Obecnou rovnici kvadriky můžeme zapsat a vytvořit užitím matice kvadriky K :

```
> X:=Vector[row]([x,y,z,1]);
> K:=Matrix(a,1..4,1..4,shape=symmetric);
K := 
$$\begin{bmatrix} a(1,1) & a(1,2) & a(1,3) & a(1,4) \\ a(1,2) & a(2,2) & a(2,3) & a(2,4) \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ a(1,4) & a(2,4) & a(3,4) & a(4,4) \end{bmatrix}$$

> Kvadrika:=sort(expand(X.K.Transpose(X)), [x,y,z])=0;
```

$$Kvadrika := a(1,1)x^2 + 2a(1,2)xy + 2a(1,3)xz + a(2,2)y^2 + 2a(2,3)yz + a(3,3)z^2 + 2a(1,4)x + 2a(2,4)y + 2a(3,4)z + a(4,4) = 0$$

Rovnice kvadriky dle zadání:

```
> RovKv:=7*x^2+6*y^2+5*z^2-4*x*y-4*y*z-22*x+24*y+2*z+30=0;
```

Hodnoty koeficientů rovnice dané kvadriky, potřebné pro vytvoření její matice, získáme porovnáním obecného tvaru rovnice kvadriky *Kvadrika* s danou konkrétní rovnicí *RovKv*. To vede na soustavu jednoduchých rovnic *SoustRovKoeff*, z nichž každá má jako neznámou jeden z koeficientů $a(1,1)$, ..., $a(4,4)$ (viz následující řádky kódu).

Poznámka: Mohli jsme také určit každý koeficient zvlášť opakovaným použitím funkce *coeff* na levou stranu rovnice *RovKv*.

```
> SoustRovKoeff:={coeffs(lhs(collect(RovKv-Kvadrika,
[x,y,z],distributed)))};
> KoeffRovKv:=solve(SoustRovKoeff,{a(1,1),a(1,2),a(1,3),
a(1,4),a(2,2),a(3,3),a(2,3),a(2,4),a(3,4),a(4,4)});
```

Rovnosti $a(i,j)=\text{číslo}$ převedeme na přiřazovací příkazy užitím funkce *assign*.

```
> assign(KoeffRovKv);
```

Tím se do obecného tvaru matice kvadriky *K* dosadí konkrétní hodnoty. Matice *K* dané kvadriky má pak tvar:

```
> K;
```

$$\begin{bmatrix} 7 & -2 & 0 & -11 \\ -2 & 6 & -2 & 12 \\ 0 & -2 & 5 & 1 \\ -11 & 12 & 1 & 30 \end{bmatrix}$$

Diskriminant kvadriky Δ

```
> Delta:=det(K);
```

$$\Delta := -972$$

Diskriminant je různý od nuly. Kvadrika je tedy **regulární**.

Hlavní minor kvadriky A_{44}

```
> SubK:=submatrix(K,1..3,1..3);
```

$$\text{SubK} := \begin{bmatrix} 7 & -2 & 0 \\ -2 & 6 & -2 \\ 0 & -2 & 5 \end{bmatrix}$$

```
> A44:=det(SubK);
```

$$A_{44} := 162$$

Hlavní minor kvadriky A_{44} je různý od nuly. Kvadrika je tedy **středová**.

Charakteristická rovnice kvadriky

Nejprve vytvoříme jednotkovou matici *E*:

```
> E:=evalm(array(1..3,1..3,identity));
```

$$E := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Potom definujeme matici **ChM** příslušné homogenní soustavy, která vede k charakteristické rovnici **ChR**:

```
> ChM:=evalm(SubK-lambda*E);
```

$$ChM := \begin{bmatrix} 7 - \lambda & -2 & 0 \\ -2 & 6 - \lambda & -2 \\ 0 & -2 & 5 - \lambda \end{bmatrix}$$

```
> ChR:=det(ChM)=0;
```

$$ChR := 162 - 99\lambda + 18\lambda^2 - \lambda^3 = 0$$

Kořeny charakteristické rovnice jsou **vlastní čísla kvadriky**:

```
> ResChR:=solve(ChR,lambda);
```

$$ResChR := 3, 6, 9$$

Pro snazší manipulaci můžeme vlastní čísla zapsat jako složky vektoru (uspořádané trojice) λ :

```
> lambda:=[ResChR];
```

$$\lambda := [3, 6, 9]$$

Potom ke konkrétnímu vlastnímu číslu přistoupíme prostřednictvím odpovídajícího indexu (pořadového čísla v uspořádané trojici):

```
> lambda[1]; lambda[2]; lambda[3];
```

3

6

9

Kanonický tvar rovnice kvadriky

Absolutní člen rovnice $\frac{\Delta}{A_{44}}$:

```
> Delta/A44;
```

-6

```
> kr:=lambda[1]*x^2+lambda[2]*y^2+9*z^2=-Delta/A44;
```

$$kr := 3x^2 + 6y^2 + 9z^2 = 6$$

Konečná podoba kanonického tvaru rovnice dané kvadriky:

```
> KanRovKv:=kr/(abs(Delta/A44));
```

$$KanRovKv := \frac{x^2}{2} + y^2 + \frac{3z^2}{2} = 1$$

Danou kvadrikou je trojosý elipsoid.

Délky poloos elipsoidu:

```
> a:=sqrt(1/coeff(lhs(KanRovKv),x,2));
> b:=sqrt(1/coeff(lhs(KanRovKv),y,2));
> c:=sqrt(1/coeff(lhs(KanRovKv),z,2));
```


$$a = \sqrt{2}$$

$$b = 1$$

$$c = \frac{\sqrt{6}}{3}$$

Zobrazení elipsoidu v transformované soustavě souřadnic provedeme pomocí funkce `implicitplot`

```
> implicitplot3d(KanRovKv,x=-2..2,y=-2..2,z=-2..2,grid=[30,30,30],
style=patchcontour,orientation=[40,55],axes=frame,tickmarks=[3,3,3],
color=COLOR(RGB,250/255,250/255,250/255),light=[90,-5,1,1,1]);
```



Vyšetření polohy kvadriky v původní soustavě souřadnic

Souřadnice středu kvadriky.

Uvažujme vzájemnou polohu přímky *Primka* a dané kvadriky *RovKv*:

```
> Primka:=[x=m+t*u,y=n+t*v,z=p+t*w];
> RovKv;
```

$$7x^2 + 6y^2 + 5z^2 - 4xy - 4yz - 22x + 24y + 2z + 30 = 0$$

Dosazení parametrických rovnic přímky za x , y , a z do rovnice kvadriky vede k následující rovnici s proměnnou t :

```
> RovKvPr:=simplify(eval(RovKv,Primka));
```

$$\begin{aligned} \text{RovKvPr} := & 7m^2 + 14mtu + 7t^2u^2 + 6n^2 + 12ntv + 6t^2v^2 + 5p^2 + 10ptw \\ & + 5t^2w^2 - 4mn - 4mtv - 4tun - 4t^2uv - 4np - 4ntw - 4tvp \\ & - 4t^2vw - 22m - 22tu + 24n + 24tv + 2p + 2tw + 30 = 0 \end{aligned}$$

Rovnici můžeme zapsat ve tvaru $At^2 + Bt + C = 0$, kde koeficienty A, B, C mají následující tvar:

```
> A:=coeff(lhs(RovKvPr),t^2);
```

$$A := 7u^2 + 6v^2 + 5w^2 - 4uv - 4vw$$

```
> B:=1/2*coeff(lhs(RovKvPr),t);
```

$$B := 7mu + 6nv + 5pw - 2mv - 2un - 2nw - 2vp - 11u + 12v + w$$

```
> C:=coeff(lhs(RovKvPr),t,0);
```

$$C := 7m^2 + 6n^2 + 5p^2 - 4mn - 4np - 22m + 24n + 2p + 30$$

Středem kvadriky je bod $S = [m, n, p]$, pro jehož souřadnice je koeficient B roven nule, tj. $B = 0$, bez ohledu na souřadnice $[u, v, w]$ směrového vektoru přímky.

```
> B1:=collect(B,[u,v,w]);
```

$$B1 := (7m - 11 - 2n)u + (6n - 2m + 12 - 2p)v + (-2n + 5p + 1)w$$

Souřadnice středu kvadriky tak určíme řešením následující homogenní soustavy rovnic s neznámými m, n, p :

```
> RStr:=coeffs(B1,[u,v,w]);
```

$$RStr := 6n - 2m + 12 - 2p, -2n + 5p + 1, 7m - 11 - 2n$$

```
> RStr_res:=solve({RStr},{m,n,p});
```

$$RStr_res := \{m = 1, p = -1, n = -2\}$$

Střed kvadriky:

```
> S:=eval([m,n,p],RStr_res);
```

$$S := [1, -2, -1]$$

Hlavní směry kvadriky

Řešíme příslušné homogenní soustavy rovnic, postupně pro všechny tři vlastní čísla.

1) $\lambda_1 = 3$

```
> Math1Sm1:=evalm(SubK-lambda[1]*E);
```

$$\text{MatHlSm1} := \begin{bmatrix} 4 & -2 & 0 \\ -2 & 3 & -2 \\ 0 & -2 & 2 \end{bmatrix}$$

> `RovHlSm1:=geneqns(MatHlSm1,[u,v,w]);`

$$\text{RovHlSm1} := \{-2u + 3v - 2w = 0, -2v + 2w = 0, 4u - 2v = 0\}$$

> `HlSm1:=solve(RovHlSm1,{u,v,w});`

$$\text{HlSm1} := \{v = 2u, w = 2u, u = u\}$$

> `u1:=eval([u,v,w],HlSm1);`

$$u1 := [u, 2u, 2u]$$

Hlavní směr u_1 :

> `u1:=eval(u1,u=1);`

$$u1 := [1, 2, 2]$$

2) $\lambda_2 = 6$

> `MatHlSm2:=evalm(SubK-lambda[2]*E);`

$$\text{MatHlSm2} := \begin{bmatrix} 1 & -2 & 0 \\ -2 & 0 & -2 \\ 0 & -2 & -1 \end{bmatrix}$$

> `RovHlSm2:=geneqns(MatHlSm2,[u,v,w]);`

$$\text{RovHlSm2} := \{u - 2v = 0, -2v - w = 0, -2u - 2w = 0\}$$

> `HlSm2:=solve(RovHlSm2,{u,v,w});`

$$\text{HlSm2} := \{u = 2v, w = -2v, v = v\}$$

> `u2:=eval([u,v,w],HlSm2);`

$$u2 := [2v, v, -2v]$$

Hlavní směr u_2 :

> `u2:=eval(u2,v=1);`

$$u2 := [2, 1, -2]$$

3) $\lambda_3 = 9$

> `MatHlSm3:=evalm(SubK-lambda[3]*E);`

$$\text{MatHlSm3} := \begin{bmatrix} -2 & -2 & 0 \\ -2 & -3 & -2 \\ 0 & -2 & -4 \end{bmatrix}$$

```

> RovHlSm3:=geneqns(MathHlSm3,[u,v,w]);
RovHlSm3 := {-2u - 2v = 0, -2u - 3v - 2w = 0, -2v - 4w = 0}
> HlSm3:=solve(RovHlSm3,{u,v,w});
HlSm3 := {u = 2w, v = -2w, w = w}
> u3:=eval([u,v,w],HlSm3);
u3 := [2w, -2w, w]

```

Hlavní směr u_3 :

```

> u3:=eval(u3,w=1);
u3 := [2, -2, 1]

```

Hlavní roviny

Odvodíme obecnou rovnici průměrové roviny sdružené se směrem $[u, v, w]$

```

> U:=Vector[row]([u,v,w,0]);
U := [u, v, w, 0]
> PrumerR:=collect(expand(evalm(U*K&K*Transpose(X))),[x,y,z])=0;

```

$$PrumerR := (7u - 2v)x + (-2u + 6v - 2w)y + (-2v + 5w)z - 11u + 12v + w = 0$$

Postupným dosazením hlavních směrů dostaneme obecné rovnice příslušných hlavních rovin:

```

> HlR1:=eval(PrumerR,[u=u1[1],v=u1[2],w=u1[3]]);
> HlR2:=eval(PrumerR,[u=u2[1],v=u2[2],w=u2[3]]);
> HlR3:=eval(PrumerR,[u=u3[1],v=u3[2],w=u3[3]]);

```

$$\begin{aligned}
 HlR1 &:= 3x + 6y + 6z + 15 = 0 \\
 HlR2 &:= 12x + 6y - 12z - 12 = 0 \\
 HlR3 &:= 18x - 18y + 9z - 45 = 0
 \end{aligned}$$

Zobrazení kvadriky v původní poloze spolu s jejími osami a hlavními rovinami

Můžeme si definovat barvu(y) pro obarvení grafu(ů):

```

> col1:=COLOR(RGB,250/255,250/255,250/255):

```

Jednotlivé grafy uložíme do proměnných:

1) Kvadrika:

```

> Kvg:=implicitplot3d(RovKv,x=-2..3,y=-4..0,z=-3..2,axes=frame,
color=col1,style=patchnogrid,grid=[40,40,40],light=[60,20,1,1,1],
tickmarks=[3,3,3],orientation=[52,63],scaling=constrained):

```

2) Osy elipsoidu:


```
> o1g:=plot3d(evalm(S+t*u1),t=-2..2,j=-1..1,thickness=3):
> o2g:=plot3d(evalm(S+t*u2),t=-2..2,j=-1..1,thickness=3):
> o3g:=plot3d(evalm(S+t*u3),t=-2..2,j=-1..1,thickness=3):
```

3) Hlavní roviny:

```
> H1R1g:=plot3d(solve(H1R1,z),x=-2..3,y=-4..0,color=grey,
style=patchnogrid,contours=60):
> H1R2g:=plot3d(solve(H1R2,z),x=-2..3,y=-4..0,color=red,
style=patchnogrid,contours=60):
> H1R3g:=plot3d(solve(H1R3,z),x=-2..3,y=-4..0,color=green,
style=patchnogrid,contours=60):
```

K zobrazení více grafů v jedné soustavě použijeme příkaz `plots[display]`:

```
> display(Kvg,o1g,o2g,o3g,H1R1g,H1R2g,H1R3g,axes=frame,
scaling=constrained,orientation=[143,75],view=[-2..3,-4..0,-3..2]);
```



Závěr

V této publikaci jsme se seznámili se základy teorie kvadrik, která navazuje na teorii kuželoseček. Po úvodní teoretické části, která se zabývá obecnými vlastnostmi kvadratických ploch, jsou ve druhé části studovány vlastnosti jednotlivých kvadrik se zřetelem k jejich použití v praxi.

Svémi jedinečnými vlastnostmi jsou kvadriky nedílnou součástí našeho života.